

TD 1 : Etudes de suites (Corrige)

Jean-Sebastien ROY, Septembre 1998

Ex 1 : Les poissons

[Avant tout travail, toujours bien penser a faire un `restart`. Idem si rien ne va plus.

[`> restart;`

[On definit la fonction qui a $u(n)$ associe $u(n+1)$:

[`> f := x -> R*x;`

$$f := x \rightarrow Rx$$

[Puis la definition recursive de la suite :

[`> u := n -> f(u(n-1));`

$$u := n \rightarrow f(u(n-1))$$

[La valeur initiale de la suite :

(on la choisit flottante (i.e. a virgule) pour que Maple n'effectue pas les calculs exacts, ce qui prendrait un temps considerable)

[`> u(0):=2.0;`

$$u(0) := 2$$

[Et la valeur du parametre R :

[`> R:=1.5;`

$$R := 1.5$$

[Calcul des premiers termes :


[`> seq(u(i), i=1..10);`

3.0, 4.50, 6.750, 10.1250, 15.18750, 22.781250, 34.1718750, 51.25781250,
76.88671875, 115.3300781

[Trace de l'evolution de la suite (bien noter la position des crochets)

[On trace la sequence des couples $[i, u(i)]$:

[`> plot([seq([i, u(i)], i=1..10)], style=point);`



[Expression non-recursive de la suite :

(L'exemple traite ici est trivial. Maple sait traiter les recurrences lineaires a coefficients constants et certains autres cas. Voir l'aide de `rsolve` : `?rsolve`)

[On veut résoudre quelque soit R , donc on efface sa valeur ainsi :

```
[ > R := 'R' ;
```

$$R := R$$

[Puis on résout la récurrence. Bien noter que l'on n'utilise pas u mais v , qui n'est pas encore défini.
En effet, Maple chercherait sinon à calculer $u(n)$, et n n'étant pas défini, la récurrence ne s'arrêterait jamais.

```
[ > e := rsolve(v(n+1)=R*v(n), v(n));
```

$$e := v(0) R^n$$

[Démonstration que $u(n)$ ne convenait pas :

```
[ > u(n);
```

Error, (in u) too many levels of recursion

[On aimerait maintenant définir la suite $v(n)$, pour cela il faut transformer l'expression e en une fonction.

C'est l'objet de la fonction `unapply(expression, variable)` :

```
[ > v := unapply(e, n);
```

$$v := n \rightarrow v(0) R^n$$

[Comme pour $u(n)$, on définit la valeur initiale :

```
[ > v(0) := 2;
```

$$v(0) := 2$$

[Et la valeur du paramètre :

```
[ > R := 1.5;
```

$$R := 1.5$$

[Des lors :

```
[ > v(n);
```

$$2 \cdot 1.5^n$$

[Limite de la suite :

```
[ > limit(v(n), n=infinity);
```

$$\infty$$

Ex 2 : Trop de poissons nuit aux poissons

[Meme principe :

```
[ > restart;
```

```
[ > f := x -> R*x*(1-x/100000);
```

$$f := x \rightarrow R x \left(1 - \frac{1}{100000} x \right)$$

```
[ > u := n -> f(u(n-1));
```

$$u := n \rightarrow f(u(n-1))$$

```
[ > R := 1.5;
```

$$R := 1.5$$

```
[ > u(0) := 2.0;
```

$$u(0) := 2.0$$

[On aurait pu se passer de la fonction f . Malheureusement, la définition de $u(n)$ aurait fait intervenir deux fois le calcul de $u(n-1)$. Ainsi pour calculer $u(10)$, Maple aurait eu à

calculer deux fois $u(9)$, quatre fois $u(8)$, huit fois $u(7)$... et le temps de calcul aurait rapidement été prohibitif. Avec la définition précédente, pour calculer $u(n)$, Maple calcule une fois $u(n-1)$, et applique f .

On verra plus loin une autre méthode pour circonvenir au problème.

```
> seq(u(i), i=1..10);
```

```
2.999940000, 4.499775005, 6.749358788, 10.12335487, 15.18349508, 22.77178454,  
34.14989850, 51.20735452, 76.77169889, 115.0691399
```

Traçons l'escalier :

La fonction `marche` renvoie les extrémités d'un petit segment vertical sous forme d'une séquence de deux couples : $[u(i), u(i)]$ (début) et $[u(i), u(i+1)]$ (fin).

```
> marche:= i -> ([u(i),u(i)], [u(i),u(i+1)]):
```


En traçant plusieurs `marche` on crée l'escalier. (bien noter la position des crochets, comme toujours)

```
> seq(marche(i), i=0..2);
```

```
[2.0, 2.0], [2.0, 2.999940000], [2.999940000, 2.999940000],  
[2.999940000, 4.499775005], [4.499775005, 4.499775005],  
[4.499775005, 6.749358788]
```

Les traits horizontaux apparaissent car Maple fait un trait entre chaque point.

```
> plot(%);
```



En nommant chacun des graphes (avec un ':' à la fin des expressions pour éviter l'horrible charabia de maple), et en utilisant `plots[display]` on arrive à superposer des graphes.


```
> p1:=plot([seq(marche(i), i=0..40)]):
```

Avec $[f(x), x]$ on trace simultanément $y=f(x)$ et $y=x$.

```
> p2:=plot([f(x), x], x=0..100000, color=[green, blue]):
```

L'option `scaling=constrained` donne des axes orthonormés :

```
> plots[display]({p1, p2}, scaling=constrained);
```



[Experimentons diverses valeurs de R :


> **R:=2.6;**

R := 2.6

[> **p1:=plot([seq(marche(i),i=0..20)]):**

[> **p2:=plot([f(x),x],x=0..100000,color=[green,blue]):**

[> **plots[display]({p1,p2},scaling=constrained);**



[La suite converge encore. Essayons une valeur plus elevee pour R:


> **R:=3.25;**

R := 3.25


[> **p1:=plot([seq(marche(i),i=0..20)]):**

[> **p2:=plot([f(x),x],x=0..100000,color=[green,blue]):**

[> **plots[display]({p1,p2},scaling=constrained);**



[La suite ne converge plus. En fait elle oscille entre deux valeurs :
 [> `p1:=plot([seq(marche(i),i=20..30)])`;
 [> `plots[display]({p1,p2},scaling=constrained)`;



[Pourquoi le comportement a-t-il change ?
 [Cherchons les points attractifs de f :
 [Quelque soit R ...

[> `R:='R'` ;

$R := R$

[On resoud $f(x)=x$: (point fixe)
 [> `s:=[solve(f(x)=x,x)]`;

$$s := \left[0, 100000 \frac{R-1}{R} \right]$$

La fonction derive de f s'ecrit :

> **D(f);**

$$x \rightarrow R \left(1 - \frac{1}{100000} x \right) - \frac{1}{100000} R x$$

Ainsi :

> **D(f)(s[2]);**

$$R \left(1 - \frac{R-1}{R} \right) - R + 1$$

Pour quelles valeurs de R le point fixe est-il attractif ?

(" ou % pour Maple VR5 signifie 'reprendre le resultat du calcul precedent)

> **solve(abs(%)<1,R);**

RealRange(Open(1), Open(3))

Reponse : pour R dans]1, 3[.

Mais pourquoi, pour R=3.25, la suite oscillait-elle entre deux valeurs ?

Pour cela il faut etudier le comportement des suites $u(2*n)$ et $u(2*n+1)$ ce qui correspond a etudier $f \circ f$.

> **R:=3.25;**


R := 3.25

@@ est l'operateur de composition repete : $f@@2$ signifie $f \circ f$.

> **p2:=plot([f(x), x, (f@@2)(x)], x=0..100000, color=[green, blue, orange]);**

> **p3:=plot([seq(marche(i), i=20..30)]):**

> **plots[display]({p3,p2}, scaling=constrained);**




On remarque deux points fixes. Pour quelles valeurs de $R>3$ sont-ils attractifs ?

> **R:='R';**

```

R:=R
> s:=[solve((f@@2)(x)=x,x)];
s:= $\left[0, 100000 \frac{R-1}{R}, \frac{50000 R+50000+50000 \sqrt{R^2-2 R-3}}{R}, \frac{50000 R+50000-50000 \sqrt{R^2-2 R-3}}{R}\right]$ 
> simplify(D(f@@2)(s[3])-D(f@@2)(s[4]));
0
... les deux valeurs des derivees de f o f sont identiques pour les deux points fixes. On
n'en etudie qu'une.
> plot([D(f@@2)(s[3]),1,-1],R=3..5,color=[blue,black,black])
;
```



```

> solve(D(f@@2)(s[3])=1,R);
3, -1
> solve(D(f@@2)(s[3])=-1,R);
1+√6, 1-√6
> evalf(%);
3.449489743, -1.449489743
```


Conclusion : jusqu'à $R=3.45$, il y a deux points fixes attractifs.

On choisit une nouvelle valeur pour R.

```

> R:=3.5;
R:=3.5
> p2:=plot([f(x),x,(f@@4)(x)],x=0..100000,color=[green,blue,
orange]);
> p3:=plot([seq(marche(i),i=50..90)]):
```

```
> plots[display]({p3,p2}, scaling=constrained);
```



[On pourrait continuer l'étude sur le même modèle pendant longtemps.


```
> R:=4.1;
```

$R := 4.1$

```
> p2:=plot([f(x),x], x=-10000..110000, color=[green,blue]):
```

```
> p3:=plot([seq(marche(i), i=0..13)]):
```

```
> plots[display]({p3,p2}, scaling=constrained);
```



[Pour $R > 4$, apparemment, la suite diverge.

Ex 3 : Des godzillas et des poissons

[> `restart;`
 [> `R:=1.1;`

`R := 1.1`

Cette fois, les suites sont récursives et font appel l'une à l'autre. Si Maple ne se souvient pas des valeurs précédemment calculées, le calcul des valeurs de la suite sera très long. Pour éviter cela, on utilise l'option `remember`, qui force Maple à stocker toutes les valeurs qu'il calcule, de façon à ne pas les calculer deux fois.

`max(0, . . . )` impose à la population d'être positive ou nulle.

[> `u:= proc(n) option remember;`
 `max(0,R*u(n-1)*(1-u(n-1)/100000-v(n-1)/5000));`
 `end;`

[> `u(0):=4200.0;`

`u(0) := 4200.0`


[> `v:= proc(n) option remember;`
 `max(0,v(n-1)*(1-v(n-1)/25000+min(4,u(n-1)/1000))/4);`
 `end;`

[> `v(0):=1.0;`

`v(0) := 1.0`

[> `p1:=plot([seq([i,u(i)],i=0..500)],color=red):`
 [> `p2:=plot([seq([i,v(i)],i=0..500)],color=green):`

```
> plots[display]({p1,p2});
```



[De manière assez étonnante, les populations suivent un cycle d'environ 40 ans.

[Changeons la valeur de R . Comme Maple se souvient de toutes les valeurs précédemment calculées, il faut les lui faire oublier, et lui redonner les valeurs initiales.

```
> R:=2.5;
```

$R := 2.5$

```
[ > readlib(forget):
```


```
[ > forget(u);u(0):=4200.0:
```

```
[ > forget(v);v(0):=1.0:
```

```
[ > p1:=plot([seq([i,u(i)],i=0..50)],color=red):
```

```
[ > p2:=plot([seq([i,v(i)],i=0..50)],color=green):
```


```
> plots[display]({p1,p2});
```



La population des poissons croit très rapidement ($R=2.5$). Malheureusement, ceci finit par faire croître le nombre de godzillas de manière très importante, ceux-ci mangent tous les poissons, puis s'éteignent. Il suffit de peu de choses pour rompre l'équilibre.

Ex 4 : Suite de fonctions

```
[ > restart;
[ Etudions la convergence simple de :
[ > f:= (x,n) -> n*cos(x)^n*sin(x);
[                                      $f := (x, n) \rightarrow n \cos(x)^n \sin(x)$ 
[ Sur [0,Pi/2]
[ (assume permet d'imposer une contrainte a une variable. additionally permet
[ d'ajouter une nouvelle contrainte sans effacer la première)
[ > assume(x>0);additionally(x<Pi/2);
[ > limit(f(x,n),n=infinity);
[                                     0
[ La suite de fonction converge simplement vers la fonction nulle.
[ Trace des graphes :
[ (red$10 equivaut a red, red, ..., red (dix fois))
[ > plot([seq(f(x,n),n=1..10)],x=0..Pi/2,scaling=constrained,c
[   olor=[red$10]);
```




On s'intéresse à la convergence uniforme sur $[a, \pi/2]$, $a > 0$.

Étudions la dérivée de f :

```
> g:=D[1](f);
```

$$g := (x, n) \rightarrow -\frac{n^2 \cos(x)^n \sin(x)^2}{\cos(x)} + n \cos(x)^n \cos(x)$$

```
> plot([seq(g(x,n), n=1..10)], x=0..Pi/2, color=[red$10]);
```



Pour quelles valeurs de x , g , la dérivée de f , s'annule-t-elle ?

```
> s:=[solve(g(x,n)=0,x)];
```

$$s := \left[\arctan\left(\frac{1}{\sqrt{n}}\right), -\arctan\left(\frac{1}{\sqrt{n}}\right) \right]$$

```
> x0:=unapply(s[1],n);
```

$$x_0 := n \rightarrow \arctan\left(\frac{1}{\sqrt{n}}\right)$$

> g(0,n);

$$n$$

> f(0,n);

$$0$$

> f(Pi/2,n);

$$0$$

Ainsi, g s'annule une fois en x_0 entre 0 et $\pi/2$, est positive avant $x_0(n)$ (donc f est croissante entre 0 et $x_0(n)$), et comme $f(0)=f(\pi/2)=0$, et g est de signe constant sur $x_0(n) \dots \pi/2$, f est necessairement decroissante sur $x_0(n) \dots \pi/2$. (toutes les fonctions manipulees sont C1)

Etudions x_0 :

> D(x0)(n);

$$-\frac{1}{2} \frac{1}{n^{3/2} \left(\frac{1}{n} + 1 \right)}$$

... manifestement negatif, donc x_0 est decroissante et :

> limit(x0(n),n=infinity);

$$0$$

Donc, a partir d'un certain rang, $x_0 < a$ et f atteint son max en a, qui tends vers 0 (convergence simple, voir plus haut).
Donc f CU vers 0 sur $[a, \pi/2]$.

Convergence uniforme sur $[0, \pi/2]$:

Sur $[0, \pi/2]$, f atteint son max en x_0 :

> f(x0(n),n);

$$\frac{\sqrt{n} \left(\frac{1}{\sqrt{\frac{1}{n} + 1}} \right)^n}{\sqrt{\frac{1}{n} + 1}}$$

> limit(%,n=infinity);

$$\infty$$

D'où la non convergence uniforme.

Calcul des integrales :

> x:='x':

> int(f(x,n),x=0..Pi/2);

$$\lim_{x \rightarrow (1/2)\pi^-} -\frac{\cos(x)^{(1+n)} n}{1+n} + \frac{n}{1+n}$$

Pas de chance, Maple n'arrive pas a calculer cette limite pourtant fort simple.
Avec op on extrait un morceau de l'expression :

> `op(1,%);`

$$-\frac{\cos(x)^{(1+n)} n}{1+n} + \frac{n}{1+n}$$

> `subs(x=Pi/2,%);`

$$-\frac{\cos\left(\frac{1}{2}\pi\right)^{(1+n)} n}{1+n} + \frac{n}{1+n}$$

> `simplify(%);`

$$\frac{n}{1+n}$$

Manifestement different de l'integrale de la limite simple (0) : on ne peut pas echanger limite et integrale.

Ex 5 : Recurrence a solution polynomiale

> `restart;`

On definit l'equation verifiee par la suite :

> `eq1:=n*u(n+2)-5*u(n+1)-(n+1)*u(n);`

$$eq1 := n u(n+2) - 5 u(n+1) - (n+1) u(n)$$

Puis un polynome quelconque de degre trois :

(L'opérateur '.' concatene les elements autour de lui)

> `a . 2;`

$$a2$$

> `add((a.i)*n^i,i=0..3);`

$$a0 + a1 n + a2 n^2 + a3 n^3$$

> `p:=unapply(%,n);`

$$p := n \rightarrow a0 + a1 n + a2 n^2 + a3 n^3$$

Ne pas se tromper sur l'ordre des parametres de la fonction subs.

Ici on substitue une fonction (u) par une autre (p).

> `subs(u=p,eq1);`

$$n p(n+2) - 5 p(n+1) - (n+1) p(n)$$

On regroupe les termes en n^qqchose

> `collect(%,n);`

$$(-2 a2 - 3 a3) n^2 + (-4 a1 - 7 a3 - 6 a2) n - 6 a0 - 5 a3 - 5 a2 - 5 a1$$

Attention : si on avait divise eq1 par n, coeffs ne marcherait pas car l'expresion ne serait pas vraiment un polynome.

On extrait les coefficients :

> `coeffs(%,n);`

$$-4 a1 - 7 a3 - 6 a2, -6 a0 - 5 a3 - 5 a2 - 5 a1, -2 a2 - 3 a3$$

Qui doivent tous etre nuls si l'equation est verifiee :

> `solve({%});`

```

[ { a3 = 2 a1, a0 = 0, a2 = -3 a1, a1 = a1 }
[ D'ou l'ensemble des polynomes possibles :
[ > subs(%, p(n));
[
[  $a1 n - 3 a1 n^2 + 2 a1 n^3$ 
[ Verification :
[ > u:=unapply(%, n);
[
[  $u := n \rightarrow a1 n - 3 a1 n^2 + 2 a1 n^3$ 
[ > simplify(eq1);
[
[ 0

```

Ex 6 : Developpements limites

On cherche un polynome unitaire quelconque de degre 3 (manifestement necessaire pour que la suite converge)

```

[ > restart;
[ > P:=n*n*n+a*n*n+b*n+c;
[
[  $P := n^3 + a n^2 + b n + c$ 

```

L'expression de la suite :

```

[ > E:=P^(1/3)-(n^4+3*n^2)^(1/4);
[
[  $E := (n^3 + a n^2 + b n + c)^{1/3} - (n^4 + 3 n^2)^{1/4}$ 

```

Que l'on developpe en serie au voisinage de plus l'infini :

```

[ > S:=series(E,n=infinity,2);
[
[
[  $S := \frac{1}{3} a + \frac{\frac{1}{3} b - \frac{1}{9} a^2 - \frac{3}{4}}{n} + O\left(\frac{1}{n^2}\right)$ 

```

On supprime le $O(1/n^2)$, qui converge absolument.

Le reste doit etre nul :

```

[ > SP:=convert(S, polynom);
[
[  $SP := \frac{1}{3} a + \frac{\frac{1}{3} b - \frac{1}{9} a^2 - \frac{3}{4}}{n}$ 
[ > F:=subs(n=1/m, SP);
[
[  $F := \frac{1}{3} a + \left(\frac{1}{3} b - \frac{1}{9} a^2 - \frac{3}{4}\right) m$ 
[ > collect(F, m);
[
[  $\frac{1}{3} a + \left(\frac{1}{3} b - \frac{1}{9} a^2 - \frac{3}{4}\right) m$ 
[ > C:={coeffs(%, m)};
[
[  $C := \left\{\frac{1}{3} a, \frac{1}{3} b - \frac{1}{9} a^2 - \frac{3}{4}\right\}$ 
[ > solve(C, {a, b});

```

$$\{a = 0, b = \frac{9}{4}\}$$

[assign permet d'affecter aux variables les valeurs trouvees par solve :

[> **assign(%)**;

[Les polynomes convenant sont de la forme :

[> **P**;

$$n^3 + \frac{9}{4}n + c$$